

```
/*
 * Simple in-line USB cable light switch controller, Erik Henneberg - April 2024
 * uses an Arduino Nano without active parts, e.g. voltage conv., serial conv.
 * and a GY-302 pcb with a BH1750 lux sensor plus a IRLD120 MOS-FET as switch.
 * The ATmega328p idle sleep works for 8 sec in between lux sensor readings.
 * The power consumption is approx. 2.5 mA.
 */
#include <Wire.h>
#include <BH1750.h>
#include "LowPower.h"

#define switch_pin A0    // A0 (PC0) as output , active high for switch
#define luxOn 2          // lux level for switch ON
#define luxOff 55555     // lux level for switch OFF

BH1750 lightMeter;

void setup() {
  pinMode(switch_pin, OUTPUT); // switch pin as output
  Wire.begin();               // start i2c communication
  lightMeter.begin();
}

void loop() {
  // read Lux and set the smart plug output accordingly
  float lux = lightMeter.readLightLevel();
  unsigned int intLux = floor(lux);
  if ((intLux > luxOn) && (intLux < luxOff)) {
    digitalWrite(switch_pin, 1); // set switch ON
  }
  else { // current light level at the min. or max. settings for output ON ?
    if ((intLux == 0) && (luxOn == 0)) {
      digitalWrite(switch_pin, 1); // set switch ON
    }
    else {
      if ((intLux < luxOn) || (intLux > luxOff)) {
        digitalWrite(switch_pin, 0); // disable switch = OFF
      }
      else { // current light level at the min. or max. settings for output OFF ?
        if ((intLux == 0) && (luxOff == 0)) {
          digitalWrite(switch_pin, 0); // disable switch = OFF
        }
      }
    }
  }
  LowPower.idle(SLEEP_8S, ADC_OFF, TIMER2_OFF, TIMER1_OFF, TIMER0_OFF,
               SPI_OFF, USART0_OFF, TWI_ON);
}
```